Lecture 07 Control Flow Instructions

CS213 – Intro to Computer Systems Branden Ghena – Fall 2023

Slides adapted from:

St-Amour, Hardavellas, Bustamente (Northwestern), Bryant, O'Hallaron (CMU), Garcia, Weaver (UC Berkeley)

Get started on Bomb Lab right away

- Bomb lab available now
 - What should you do before the exam?
 - Phases 1-3 of Bomb Lab
 - They are good practice for the kinds of assembly problems I'll put on the exam
 - Phases 4-6 are harder and can honestly wait
 - We'll talk about stuff in lectures on "Procedures" and "Pointers, Arrays, and Structs" that will help with this part

Today's Goals

- Understand converting C control flow statements to assembly
 - If, If-else, While, For, etc.
- Discuss multiple ways to represent code
 - Often an efficiency tradeoff

Outline

Condition Codes

• Branching (If/Else)

Loops (Do While, While, For)

Conditional Move

Condition codes

- Control is mediated via Condition codes
 - single-bit registers that record answers to questions about values
 - E.g., Is value x greater than value y? Are they equal? Is their sum even?
 - Let's keep "question" abstract for now. We'll see the details in a bit.
 - Terminology:
 - a bit is *set* if it is 1
 - a bit is *cleared* (or *reset*) if it is 0

Conditionals at the machine level

- At machine level, conditional operations are a 2-step process:
 - Perform an operation that sets or clears condition codes (ask questions)
 - Then observe which condition codes are set, do the operation (or not)
- Can express Boolean operations, conditionals, loops, etc.
 - We will see the first today, and more control next lecture
- So now we need three things:
 - 1. Instructions that compare values and set condition codes
 - 2. Instructions that observe condition codes and do something (or not)
 - 3. A set of actual condition codes (what questions do we track answers to?)

Step 1: Setting condition codes

- Analogy: Asking ALL the possible questions at once
 - And recording the answers
 - We don't know yet which question is the one we care about!
- Done in one of two ways
 - **Implicitly**: all* arithmetic instructions set (and reset) condition codes in addition to producing a result
 - *except lea; it's not "officially" an arithmetic instruction
 - **Explicitly**: by instructions whose sole purpose is to set condition codes
 - E.g., cmpq
 - They don't actually produce results (in registers or memory)
 - Condition codes are left unchanged by other operations (such as mov)

Implicitly Setting Condition Codes

- Condition codes on x86
 - **CF** Carry Flag (for unsigned) **SF** Sign Flag (for signed)
 - **ZF** Zero Flag **OF** Overflow Flag (for signed)
 - PF Parity Flag
 - Not an arbitrary set! By combining them, can keep track of answers to many useful questions! (We'll see exactly which in a bit.)

Implicitly Setting Condition Codes

```
CF (Carry) SF (Sign) ZF (Zero) OF (Overflow) PF (Parity)
```

- Set (or reset) based on the result of arithmetic operations
 Example: addq Src, Dest # C-analog: t = a+b
 - **ZF** set if t == 0
 - SF set if t < 0 (as signed encoding)
 - CF set if carry out from most significant bit (unsigned overflow)
 also CF takes the value of the last bit shifted (left or right)
 - OF set if twos-complement (signed) overflow (pos/neg overflow)
 (a>0 && b>0 && t<0) || (a<0 && b<0 && t>=0)
 also, set if a 1-bit shift operation changes the sign of the result
 - **PF set** if t has an even number of 1 bits

Explicitly Setting Condition Codes: Compare

- cmp{b,w,l,q} Src2, Src1
- cmpq src,dst computes t = dst-src, ignoring the result
 - And sets condition codes along the way, like subq would!
 - Follows the rules we saw on the previous slide for arithmetic instructions
 - Beware the order of the cmp operands!
- Use cases
 - ZF set if dst == src
 - SF set if (dst-src) < 0 (as signed), i.e., src > dst in a signed comparison!
 - CF and OF used mostly in combinations with others (see in a few slides)

Explicitly Setting Condition Codes: Test

- test{b,w,l,q} *Src2,Src1*
- testq src,dst computes t = dst&src, ignoring the result!
 - And sets condition codes like andq would (order doesn't matter here)
 - So again, same rules as arithmetic instructions
- Use cases
 - **ZF set** when dst&src == 0, i.e., a and b have no bits in common
 - SF set when dst&src < 0
- Useful when doing bit masking
 - E.g., x & 0x1, to know whether x is even or odd
 - If the result of the & is 0, it's even, if 1, it's odd

Step 2: Reading Condition Codes

- Cannot read condition codes directly; instead observe via instructions
 - And generally observe combinations of condition codes, not individual ones

- Example: the setx family of instructions
 - Write single-byte destination register based on combinations of condition codes
 - set{e, ne, s, ...} D where D is a 1-byte register
 - Example: sete %al
 - means: %al=1 if flag ZF is set, %al=0 otherwise

Using condition codes for comparison

- setle Less than or equal (signed)
 - Combination of condition codes: (SF^OF) | ZF
 - SF Sign Flag (true if negative)
 - OF Overflow Flag (true if signed overflow occurred)
 - ZF Zero Flag (true if result is zero)
- All of the combos expect to be run after a cmp src, dst
 - dst <= src (runs dst-src)
 - If:
 - The result is zero src and dst were equal
 - OR if one but not both:
 - The result is negative (and didn't overflow) src was larger than dst
 - The result overflowed (and is positive) dst is negative, src is positive

Condition codes combinations

SetX	Description	Condition
sete	Equal / Zero	ZF
setne	Not Equal / Not Zero	~ZF
sets	Negative	SF
setns	Nonnegative	~SF
setg	Greater (Signed)	~(SF^OF) &~ZF
setge	Greater or Equal (Signed)	~(SF^OF)
setl	Less (Signed)	(SF^OF)
setle	Less or Equal (Signed)	(SF^OF) ZF
seta	Above (unsigned)	~CF&~ZF
setb	Below (unsigned)	CF

Note: suffixes do not indicate operand sizes, but rather conditions

These same suffixes will come back when we see other instructions that read condition codes.

Expect to be run after a cmp

Step 2: Reading Condition Codes

- setX (and others) read the current state of condition codes
 - Whatever it is, and whichever instruction changed it last
- So when you see (for example) setne, work backwards!
 - Look at previous instructions, to find the last one to change conditions
 - Then you'll know the two values that were compared
 - Ignore instructions that don't touch condition codes (like moves)
- Usually you'll see a cmpx (or testx, or arithmetic) right before
 - But not always, so know what to do in general

What do you need to know?

- 90%+ of the time
 - cmp instruction followed by setX instruction (or a branch, next lecture)
 - Don't have to think about condition codes at all!
 - Think of as dst X src
 - $dst \le src$ **or** dst != src **etc.**
- 10% or less of the time
 - Arbitrary arithmetic instruction sets the condition codes
 - Or testq sets the condition codes
 - Followed by a setX or branch (next section)
 - And you actually have to think about which condition codes are set to figure out what the assembly is doing, which can be challenging

Break + Practice

op src, dst

- setx asks the question: "Is destination X source?"
 - Usually condition codes from "cmp source, destination"
 - Don't have to care about the exact values of the condition codes though
 - Just understand the logic

SetX	Description
sete	Equal / Zero
setne	Not Equal / Not Zero
sets	Negative
setns	Nonnegative
setg	Greater (Signed)
setge	Greater or Equal (Signed)
setl	Less (Signed)
setle	Less or Equal (Signed)
seta	Above (unsigned)
setb	Below (unsigned)

Break + Practice

 $% \sin 1 = 0 \times 01$

op src, dst

- setx asks the question: "Is destination X source?"
 - Usually condition codes from "cmp source, destination"
 - Don't have to care about the exact values of the condition codes though
 - Just understand the logic

```
%dil = 0xF0

cmp %sil, %dil

TRUE seta %al # is %dil ABOVE %sil (unsigned)

FALSE sete %al # is %dil EQUAL to %sil

FALSE setge %al # is %dil GREATER or EQUAL to %sil (signed)
```

Outline

Condition Codes

Branching (If/Else)

Loops (Do While, While, For)

Conditional Move

What can instructions do?

- Move data: √
- Arithmetic: √

Transfer control

Instead of executing next instruction, go somewhere else


```
if (x > y)
    result = x-y;
else
    result = y-x;
```

```
while (x > y)
    result = x-y;
return result;
```

- Sometimes we want to go from the red code to the green code
- But the blue code is what's next!
- Need to transfer control! Execute an instruction that is not the next one
- And conditionally, too! (i.e., based on a condition)

Breaking with sequential execution

- "Normal" execution follows instructions in listed (sequential) order
- To move to a different location jump
 - Jump to different part of code depending on condition codes
 - Destination of a jump label: particular address at which we find code
 - Label addresses are determined when generating the object code



Jumping

- jx Instructions
 - Jump to different part of code depending on condition codes
 - jmp has two options
 - **Direct**: to a label (literal address)
 - **Indirect**: based on a register
 - Direct is the most common

jΧ	Condition	Description
jmp	1	Unconditional
je	ZF	Equal / Zero
jne	~ZF	Not Equal / Not Zero
js	SF	Negative
jns	~SF	Nonnegative
jg	~(SF^OF) &~ZF	Greater (Signed)
jge	~ (SF^OF)	Greater or Equal (Signed)
j1	(SF^OF)	Less (Signed)
jle	(SF^OF) ZF	Less or Equal (Signed)
ja	~CF&~ZF	Above (unsigned)
jae	~CF	Above or Equal (unsigned)
jb	CF	Below (unsigned)
•••	• • •	

Key idea: building C constructs with assembly

- Jump will let us build the flow control statements in C
 - If, While, For, Switch, etc.
- But the translation isn't always obvious
 - Might switch ordering, or negate the logical condition
 - Maintains the same result when it runs, but easier for assembly

- Steps
 - 1. Transform C into something simpler (closer to assembly)
 - 2. Transform simpler C into assembly

The "something simpler" is goto

- C allows goto as means of transferring control
 - Closer to machine-level programming style
 - Place labels wherever you want in code
 - Goto "jumps" to the referenced label
- Generally considered bad programming style
 - Makes it really difficult to understand what code is doing

```
int i = 0;
start:
  if (i >= 3) { goto end;
  ++i;
  printf("Hello ");
  goto start;
end:
  printf("World!\n");
Prints:
"Hello Hello World!\n"
```

```
long absdiff_j(long x, long y)
 long absdiff(long x, long y)
   long result;
                                                  long result;
                                                  int ntest = (x \le y);
   if (x > y)
     result = x-y;
                                                  if (ntest) { goto Else; }
                                                  result = x-y;
   else
     result = y-x;
                                                  goto Done;
   return result;
                                              Else:
                                                  result = y-x;
                                              Done:
                                                  return result;

    Translate an if statement into
```

- Translate an if statement into a "simpler" goto statement
 - Makes the if statement closer to machine code because goto can translate to jumps

Goto Version

```
long absdiff_j(long x, long y)
    long result;
    int ntest = (x \le y);
    if (ntest) {goto Else;}
    result = x-y;
    goto Done;
Else:
    result = y-x;
Done:
    return result;
```

```
absdiff:
          %rsi, %rdi # cmp x:y
  cmpq
          .L2
                     # x <= y
  jle
          %rdi, %rax
  movq
  subq
          %rsi, %rax
  jmp
          .L3
.L2:
                     # jle target
          %rsi, %rax
  movq
  subq
          %rdi, %rax
.L3:
  ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

Goto Version

```
long absdiff_j(long x, long y)
    long result;
 \rightarrow int ntest = (x <= y);
 if (ntest) {goto Else;}
    result = x-y;
    goto Done;
Else:
    result = y-x;
Done:
    return result;
```

```
absdiff:
cmpq
         %rsi, %rdi # cmp x:y
          .L2
                    # x <= y
🛶 jle
         %rdi, %rax
  movq
  subq
         %rsi, %rax
  jmp
          .L3
.L2:
                     # jle target
          %rsi, %rax
  movq
  subq
          %rdi, %rax
.L3:
  ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

Goto Version

```
long absdiff_j(long x, long y)
    long result;
    int ntest = (x \le y);
    if (ntest) {goto Else;}

ightharpoonup result = x-y;
    goto Done;
 Else:
    result = y-x;
 Done:
    return result;
```

```
absdiff:
          %rsi, %rdi # cmp x:y
  cmpq
          .L2
                    # x <= y
  jle
          %rdi, %rax
movq
 subq
          %rsi, %rax
  jmp
          .L3
.L2:
                     # jle target
          %rsi, %rax
  movq
  subq
          %rdi, %rax
.L3:
  ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

Goto Version

```
long absdiff_j(long x, long y)
    long result;
    int ntest = (x \le y);
    if (ntest) {goto Else;}
    result = x-y;
 goto Done;
Else:
    result = y-x;
Done:
    return result;
```

```
absdiff:
          %rsi, %rdi # cmp x:y
  cmpq
          .L2
                    # x <= y
  jle
          %rdi, %rax
  movq
  subq
         %rsi, %rax
dmj
          .L3
.L2:
                     # jle target
          %rsi, %rax
  movq
  subq
          %rdi, %rax
.L3:
  ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

Goto Version

```
long absdiff_j(long x, long y)
    long result;
    int ntest = (x \le y);
    if (ntest) {goto Else;}
    result = x-y;
    goto Done;
Else:
result = y-x;
Done:
    return result;
```

```
absdiff:
         %rsi, %rdi # cmp x:y
  cmpq
         .L2
                    # x <= y
  jle
         %rdi, %rax
  movq
  subq
         %rsi, %rax
  jmp
          .L3
.L2:
                     # jle target
movq
         %rsi, %rax
subq
         %rdi, %rax
.L3:
  ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

Goto Version

```
long absdiff_j(long x, long y)
    long result;
    int ntest = (x \le y);
    if (ntest) {goto Else;}
    result = x-y;
    goto Done;
Else:
    result = y-x;
Done:
  return result;
```

```
absdiff:
          %rsi, %rdi # cmp x:y
  cmpq
          .L2
                    # x <= y
  jle
          %rdi, %rax
  movq
  subq
          %rsi, %rax
  jmp
          .L3
.L2:
                     # jle target
          %rsi, %rax
  movq
  subq
         %rdi, %rax
.L3:
ret
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

General "if-then-else" translation

C Code

```
if (test-expr)
then-statement
else
else-statement
```

- *test-expr* is an expression returning integer
 - = 0 interpreted as false, ≠0 interpreted as true
- Only one of the two statements is executed
 - i.e. only one of the two *branches* of code
- That's one translation; there are others
 - E.g., flipping the order of the blocks instead of flipping the test
- Conditional expressions (x ? y : z) can use the same translation

Goto Version

```
ntest = !(test-expr);
if (ntest) {
    goto Else;
}
then-statement;
goto done;
Else:
else-statement;
done:
```

```
long test(long a, long b) {
  long c;
  if (a > b) {
    c = 1;
  } else if (a < b) {</pre>
    c = -1;
  } else {
    c = 0;
  return c;
```

```
long test(long a, long b) {
                                     cmp %rsi, %rdi
                                     jle elif # !(a > b)
  long c;
                                     movq $1, %rax
  if (a > b) {
                                     jmp end
    c = 1;
                                   elif:
  } else if (a < b) {</pre>
                                     cmp %rsi, %rdi
    c = -1;
                                     jge else # !(a < b)</pre>
  } else {
                                     movq $-1, %rax
    c = 0;
                                     jmp end
                                   else:
  return c;
                                    movq $0, %rax
                                   end:
                                                    # returns %rax
                                     ret
```

 $a\rightarrow \$rdi, b\rightarrow \$rsi, c\rightarrow \$rax$

```
cmp %rsi, %rdi
long test(long a, long b) {
                                     jle elif # !(a > b)
  long c;
                                     movq $1, %rax
  if (a > b) {
                                     jmp end
    c = 1;
                                   elif:
  } else if (a < b) {</pre>
                                     cmp %rsi, %rdi
    c = -1;
                                     jge else # !(a < b)</pre>
  } else {
                                     movq $-1, %rax
    c = 0;
                                     jmp end
                                   else:
  return c;
                                     movq $0, %rax
                                   end:
                                                    # returns %rax
                                     ret
```

 $a\rightarrow \$rdi, b\rightarrow \$rsi, c\rightarrow \$rax$

```
long test(long a, long b) {
  long c;
  if (a > b) {
    c = 1;
  } else if (a < b) {</pre>
    c = -1;
  } else {
    c = 0;
  return c;
```

```
a\rightarrow \$rdi, b\rightarrow \$rsi, c\rightarrow \$rax
  cmp %rsi, %rdi
  jle elif # !(a > b)
  movq $1, %rax
  jmp end
elif:
  cmp %rsi, %rdi
  jge else # !(a < b)</pre>
  movq $-1, %rax
  jmp end
else:
  movq $0, %rax
end:
                    # returns %rax
  ret
```

If statement - bigger example

```
long test(long a, long b) {
  long c;
  if (a > b) {
    c = 1;
  } else if (a < b) {</pre>
    c = -1;
   else {
  return c;
```

```
a\rightarrow \$rdi, b\rightarrow \$rsi, c\rightarrow \$rax
  cmp %rsi, %rdi
  jle elif # !(a > b)
  movq $1, %rax
  jmp end
elif:
  cmp %rsi, %rdi
  jge else # !(a < b)</pre>
  movq $-1, %rax
  jmp end
else:
  movq $0, %rax
end:
                    # returns %rax
  ret
```

If statement - bigger example

```
long test(long a, long b) {
                                     cmp %rsi, %rdi
                                     jle elif # !(a > b)
  long c;
                                     movq $1, %rax
  if (a > b) {
                                     jmp end
    c = 1;
                                   elif:
  } else if (a < b) {</pre>
                                     cmp %rsi, %rdi # unnecessary
    c = -1;
                                     jge else # !(a < b)</pre>
  } else {
                                     movq $-1, %rax
    c = 0;
                                     jmp end
                                   else:
  return c;
                                     movq $0, %rax
                                   end:
                                                    # returns %rax
                                     ret
```

 $a\rightarrow \$rdi, b\rightarrow \$rsi, c\rightarrow \$rax$

```
Break + Optimization (O1)
                                        a \rightarrow rdi, b \rightarrow rsi, c \rightarrow rax
                                        movq $1, %rax
long test(long a, long b) {
                                        cmp %rsi, %rdi
  long c;
                                        jg end
  if (a > b) {
     c = 1;
  } else if (a < b) {</pre>
                                        neg %rax
     c = -1;
                                      end:
  } else {
                                        ret
                                                        # returns %rax
    c = 0;
                     What is the yellow code block doing above?
  return c;
```

```
Break + Optimization (O1)
                                         a \rightarrow \$rdi, b \rightarrow \$rsi, c \rightarrow \$rax
long test(long a, long b) {
                                          movq $1, %rax
                                          cmp %rsi, %rdi
  long c;
                                          jg end
  if (a > b) {
                                                              else if and else
     c = 1;
                                                              together
   } else if (a < b) {</pre>
                                          neg %rax
     c = -1;
                                        end:
   } else {
                                          ret
                                                           # returns %rax
     c = 0;
                      What is the yellow code block doing above?
  return c;
                             Generates 0 (not less) or -1 (less)
```

Indirect jump

- •jmp *0x40000(%rdi, %rdx, 8)
 - Calculate memory address: 0x40000 + %rdi + 8*%rdx
 - Load value from memory address
 - Jump to that value
- Indirect jumps jump to the address loaded from memory
 - Essentially a function pointer
 - Or used for a Jump Table: efficient switch statements (see bonus slides)

- The * lets you know that something tricky is going on
 - Displacement could be a label rather than a value

Outline

Condition Codes

Branching (If/Else)

Loops (Do While, While, For)

Conditional Move

Loops

- C provides different looping constructs
 - while, do ... while, for
- No corresponding instruction in machine code
- Most compilers
 - 1. Transform general loops into do ... while

```
do
body-statement
while (test-expr);
```

- 2. Rewrite that with goto
- 3. Then compile them into machine code

Do-while:

Same idea as a while loop, but the body always runs at least once

"Do-While" Loop Compilation

- Running example: count number of 1s in x ("popcount")
 - We'll write it with different kinds of loops
 - What the body of the loop does is not our focus; we'll just ignore it
- Use conditional branch to either continue looping or to exit loop

C Code

```
long pcount_do
  (unsigned long x)
{
  long result = 0;
  do {
    result += x & 0x1;
    x >>= 1;
  } while (x);
  return result;
}
```

Goto Version

```
long pcount_goto
  (unsigned long x)
{
  long result = 0;
  loop:
  result += x & 0x1;
  x >>= 1;
  if (x) {goto loop;}
  return result;
}
```

Register	Use(s)
%rdi	Argument x
%rax	result

Goto Version

```
long pcount_goto
  (unsigned long x) {
  long result = 0;

loop:
  result += x & 0x1;
  x >>= 1;
  if (x) {goto loop;}
  return result;
}
```

```
$0,%rax
                     # result = 0
   movq
▶ .L2:
                     # loop:
          %rdi,%rdx
   movq
   andq $1,%rdx
                     # t = x & 0x1
   addq %rdx,%rax # result += t
   shrq %rdi
                     # x >>= 1
          .L2
                       if (x) goto loop
   jne
   rep; ret
```

Register	Use(s)
%rdi	Argument x
%rax	result

Goto Version

```
long pcount_goto
  (unsigned long x) {
  long result = 0;

  result += x & 0x1;
  x >>= 1;
  if (x) {goto loop;}
  return result;
}
movq $0

.L2:
  movq %re
  andq $1;
  addq %re
  shrq %re
  pre
  if (x) rep; ret
```

```
movq $0,%rax # result = 0
# loop:
movq %rdi,%rdx
andq $1,%rdx # t = x & 0x1
addq %rdx,%rax # result += t
shrq %rdi # x >>= 1
jne .L2 # if (x) goto loop
rep: ret
```

Which instruction sets the condition codes for jne?

Logical shift right (shrq)

Register	Use(s)
%rdi	Argument x
%rax	result

Goto Version

```
long pcount goto
                                            $0,%rax
                                                       # result = 0
                                    movq
  (unsigned long x) {
                                 L2:
                                                         loop:
 long result = 0;
                                            %rdi,%rdx
                                    movq
loop:
                                    andq $1,%rdx
                                                       # t = x & 0x1
 result += x \& 0x1;
                                    addq %rdx,%rax # result += t
 x >>= 1;
                                    shrq
                                            %rdi
                                                       \# x >>= 1
 if (x) {goto loop;}
                                            .L2
                                                          if (x) goto loop
                                    jne
 return result;
                                    rep; ret
```

rep instruction repeats string operations following it What?!!

Register	Use(s)
%rdi	Argument x
%rax	result

Goto Version

```
long prount goto
                                                       # result = 0
                                            $0,%rax
                                    movq
  (unsigned long x) {
                                                         loop:
 long result = 0;
                                            %rdi,%rdx
                                    movq
loop:
                                    andq $1,%rdx
                                                       # t = x & 0x1
 result += x \& 0x1;
                                    addq
                                          %rdx,%rax # result += t
 x >>= 1;
                                                      # x >>= 1
                                    shrq
                                            %rdi
 if (x) {goto loop;}
                                            . L2
                                                         if (x) goto loop
                                    jne
 return result;
                                    rep; ret
```

- rep instruction repeats string operations following it What?!!
- rep; ret uses rep as a no-op (a.k.a nop, an operation that does nothing)
 - Example of a compiler optimization that you might run into in real assembly code
 - AMD recommends this to speed up execution when there is a jump before a return
 - See CE361 and CE452 for more details (Computer Architecture courses)

General "Do-While" Translation

```
• Body: {
    Statement<sub>1</sub>;
    Statement<sub>2</sub>;
    ...
    Statement<sub>n</sub>;
}

C Code

do

Body

while (Test);
```

Goto Version

```
loop:
   Body
   if (Test) {
      goto loop
   }
```

- Test returns integer
 - = 0 interpreted as false
 - ≠ 0 interpreted as true

General "While" Translation #1

- "Jump-to-middle" translation
- Most straightforward match to how "while" works

Goto Version

While version while (Test) { Body }

```
goto test;
loop:
   Body
test:
   if (Test) {
      goto loop;
   }
done:
```

While Loop Example #1

C Code

```
long pcount_while
  (unsigned long x)
{
  long result = 0;
  while (x) {
    result += x & 0x1;
    x >>= 1;
  }
  return result;
}
```

Jump to Middle

```
long pcount goto jtm
  (unsigned long x)
  long result = 0;
  goto test;
 loop:
  result += x \& 0x1;
  x >>= 1;
 test:
  if(x) {goto loop;}
  return result;
```

Initial goto starts loop at test

Comparing while to do-while

While with goto (jump to middle)

```
long prount while goto jtm
  (unsigned long x)
  long result = 0;
  goto test;
 loop:
  result += x \& 0x1;
  x >>= 1;
 test:
  if(x) {goto loop;}
  return result;
```

Do While with goto

```
long pcount dowhile goto
  (unsigned long x)
  long result = 0;
 loop:
  result += x \& 0x1;
  x >>= 1;
  if (x) {goto loop;}
  return result;
```

General "While" Translation #2

While version

while (*Test*) *Body*




Do-While Version

if (! Test)
 goto done;
do

 Body
 while(Test);
done:

- "Do-while" conversion
- More optimized compiler translation

Goto Version



"While" Loop Example #2

C Code

```
long pcount_while
  (unsigned long x)
{
  long result = 0;
  while (x) {
    result += x & 0x1;
    x >>= 1;
  }
  return result;
}
```

Goto Version

```
long pcount goto dw
  (unsigned long x)
  long result = 0;
  if (!x) {goto done;}
 loop:
 result += x \& 0x1;
 x >>= 1:
  if(x) {goto loop;}
done:
 return result;
```

Initial conditional guards entrance to loop

Comparing jump-to-middle and guarded-do-while

While with goto (jump to middle)

```
long prount while goto jtm
  (unsigned long x)
  long result = 0;
  goto test;
 loop:
  result += x \& 0x1;
  x >>= 1;
 test:
  if(x) {goto loop;}
  return result;
```

While with goto (guarded do-while)

```
long pcount goto dw
  (unsigned long x)
  long result = 0;
  if (!x) {goto done;}
 loop:
  result += x \& 0x1;
  x >>= 1;
  if(x) {goto loop;}
 done:
  return result;
```

"For" Loop Form

General Form

```
for (Init; Test; Update)

Body
```

```
#define WSIZE 8*sizeof(int)
long poount for
  (unsigned long x)
  size t i;
  long result = 0;
  for (i = 0; i < WSIZE; i++)
   unsigned bit = (x \gg i) \& 0x1;
    result += bit;
  return result;
```

Init

```
i = 0
```

Test

```
i < WSIZE
```

Update

```
i++
```

Body


```
{
  unsigned bit = (x >> i) & 0x1;
  result += bit;
}
```

"For"→ "While" → "Do-While" → "Goto"

For Version

```
for (Init; Test; Update)

Body
```




While Version

```
Init;
while (Test) {
    Body
    Update;
}
```

Goto Version

```
Init;
  if (!Test)
    goto done;
loop:
    Body
    Update;
  if (Test)
    goto loop;
done:
```





Do-While Version

```
Init;
if (!Test)
  goto done;
do {
  Body
  Update;
} while (Test)
done:
```

"For" Loop Conversion Example

C Code

```
#define WSIZE 8*sizeof(int)
long prount for (unsigned x)
  size t i;
  long result = 0;
  for (i = 0; i < WSIZE; i++) {
   unsigned bit =
       (x >> i) & 0x1;
    result += bit;
  return result;
```

Goto Version

```
#define WSIZE 8*sizeof(int)
long prount for gt(unsigned x)
  size t i;
  long result = 0;
  i = 0; Init
  if (!(i < WSIZE))
    goto done;
 loop:
                     Body
   unsigned bit =
       (x >> i) & 0x1;
    result += bit;
  i++; Update
  if (i < WSIZE)
                  Test
    goto loop;
 done:
  return result;
```

Break + Assembly to loop

What does this function do?

```
my function: # %rdi is argument1
 mov $0, %rax
 mov $0, %rbx
  test %rdi, %rdi
  je end
loop:
  add %rbx, %rax
  add $1, %rbx
  cmp %rdi, %rbx
  jne loop
end:
                  # returns %rax
  ret
```

```
my function: # %rdi is argument1
 mov $0, %rax  # clear variables
 mov $0, %rbx
  test %rdi, %rdi
                # skip loop if %rdi is 0
  je end
loop:
  add %rbx, %rax
  add $1, %rbx
  cmp %rdi, %rbx
  jne loop
end:
                 # returns %rax
  ret
```

```
my function: # %rdi is argument1
 mov $0, %rax # clear variables
 mov $0, %rbx
  test %rdi, %rdi
  je end
                 # skip loop if %rdi is 0
loop:
  add %rbx, %rax
  add $1, %rbx
  cmp %rdi, %rbx
  jne loop
end:
                 # returns %rax
  ret
```

```
my function: # %rdi is argument1
 mov $0, %rax # clear variables
 mov $0, %rbx
  test %rdi, %rdi
  je end # skip loop if %rdi is 0
loop:
  add %rbx, %rax # %rax += %rbx
 add $1, %rbx # %rbx += 1
  cmp %rdi, %rbx
  jne loop
end:
                # returns %rax
  ret
```

```
my function: # %rdi is argument1
 mov $0, %rax # clear variables
 mov $0, %rbx
  test %rdi, %rdi
                # skip loop if %rdi is 0
  je end
loop:
 add %rbx, %rax # %rax += %rbx
  add $1, %rbx # %rbx += 1
  cmp %rdi, %rbx
  jne loop
                 # while %rbx != %rdi
end:
                 # returns %rax
  ret
```

```
long my function(long rdi) {
 long rax = 0;
 long rbx = 0;
 while (rbx != rdi) {
   rax += rbx;
   rbx += 1;
 return rax;
```

```
my function: # %rdi is argument1
 mov $0, %rax # clear variables
 mov $0, %rbx
  test %rdi, %rdi
  je end
                 # skip loop if %rdi is 0
loop:
  add %rbx, %rax # %rax += %rbx
  add $1, %rbx # %rbx += 1
  cmp %rdi, %rbx
  jne loop
                 # while %rbx != %rdi
end:
                 # returns %rax
  ret
```

```
long my_function(long rdi) {
  long rax = 0;
  long rbx = 0;

while (rbx != rdi) {
   rax += rbx;
   rbx += 1;
  }

return rax;
}
```

```
long my_function(long max) {
    long result = 0;
    for (int i=0; i<max; i++) {
        result += i;
    }
    return result;
}</pre>
```

```
my function: # %rdi is argument1
 mov $0, %rax # clear variables
 mov $0, %rbx
 test %rdi, %rdi
  je end
                 # skip loop if %rdi is 0
loop:
  add %rbx, %rax # %rax += %rbx
  add $1, %rbx # %rbx += 1
  cmp %rdi, %rbx
  jne loop
                 # while %rbx != %rdi
end:
                 # returns %rax
  ret
```

Outline

Condition Codes

Branching (If/Else)

Loops (Do While, While, For)

Conditional Move

The Problem with Conditional Jumps

- Conditional jumps = conditional transfer of control
 - i.e., forget what you thought you were going to do, do this other thing instead
- Modern processors like to do work "ahead of time"
 - Keywords: pipelining, branch prediction, speculative execution
 - Transfer of control may mean throwing that work away
 - That's inefficient
- Solution: conditional moves
 - We still get to do something conditionally
 - But no transfer of control necessary
 - "Ahead of time" work can always be kept

Conditional Moves

cmovX	Description
cmove S, D	equal / Zero
cmovne S, D	not equal / Not zero
comvs S, D	negative
cmovns S, D	nonnegative
comvg S, D	greater (Signed)
cmovge S, D	greater or equal (Signed)
cmovl S, D	less (Signed)
cmovle S, D	less or equal (Signed)
cmova S, D	above (Unsigned)
cmovae S, D	above or equal (Unsigned)
cmovb S, D	below (Unsigned)
cmovbe S, D	below or equal (Unsigned)

D ← S only if test condition is true

Conditional Move Example

```
long absdiff(long x, long y)
{
  long res;
  if (x > y)
    res = x-y;
  else
    res = y-x;
  return res;
}
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	Return value

```
absdiff:
  movq %rdi, %rax # res = x
  subq %rsi, %rax # res = x-y
  movq %rsi, %rdx
  subq %rdi, %rdx # alt = y-x
  cmpq %rsi, %rdi # cmp x:y
  cmovle %rdx, %rax # if x<=y, res = alt
  ret</pre>
```

Look Ma, no branching!

Must compute both results, though, which is not always possible or desirable...

Bad Cases for Conditional Move

Expensive Computations

```
val = Test(x) ? Hard1(x) : Hard2(x);
```

- Both values get computed
- Only makes sense when computations are very simple

Risky Computations

```
val = p ? *p : 0;
```

- A cmov requires that both values get computed
- Could trigger a fault (compiler must use jumps instead)

Computations with side effects

```
val = x > 0 ? x++ : x--;
```

- Both values get computed
- Needs use extra temporary registers to hold intermediate results

```
If, else if, else – optimized (O3) a→%rdi, b→%rsi, c→%rax
long test(long a, long b) {
                                      movq $0, %rax # clear reg
                                       cmp %rsi, %rdi
  long c;
                                      movq $1, %rdx
  if (a > b) {
                                                         else if and else
                                       setl %al
     c = 1;
                                                         together
                                      neg %rax
                                                         (%al is %rax)
  } else if (a < b) {</pre>
                                       cmp %rsi, %rdi
     c = -1;
                                       cmovg %rdx, %rax
                                                         select output
  } else {
                                                      # returns %rax
                                      ret
    c = 0;
  return c;
```

Outline

Condition Codes

• Branching (If/Else)

Loops (Do While, While, For)

Conditional Move

- Bonus Slides
 - Switch Statements and Jump Tables

Switch statements

- A multi-way branching capability based on the value of an integer
- Useful when many possible outcomes
- Switch cases
 - Fall through cases:
 - Here 1
 - Missing cases:
 - Here 3, 4, 5, 6
 - Multiple case labels:
 - Here 7 & 8
- Easier to read C code and more efficient implementation with jump tables

```
long switch fun
 (long x, long y, long z, long w) {
  switch(x) {
  case 0:
       w += y;
       break;
  case 1:
       W = V;
       /* FALL THROUGH */
  case 2:
       W += Z;
       break;
  /* MISSING CASES */
  case 7:
  case 8: /* MULTIPLE CASES */
       w = z;
       break;
 default:
       w = 2;
       break;
  w += 5;
  return w;
```

Target code blocks

```
case 0:
    W += y;
    break;
case 1:
    w = v;
    /* FALL THROUGH */
case 2:
    W += Z;
    break;
case 7:
case 8: /* MULTIPLE CASES */
    w = z;
    break;
default:
    w = 2;
    break;
```

One code block per case!

```
.L7:
                     # case 0
       %rsi, %rcx
  addq
                     # break
  jmp .L2
                     # case 1
.L6:
  subq %rsi, %rcx
  # FALL THROUGH
                     # case 2
.L5:
  addq %rdx, %rcx
         .L2 # break
  jmp
.L3:
                     # cases 7 and 8
  subq
       %rdx, %rcx
         . L2
                    # break
  jmp
                     # default
.L8:
  movl $2, %ecx
                    # break
         .L2
  jmp
.L2:
```

%rdi	Argument x
%rsi	Argument y
%rdx	Argument z
%rcx	Argument w
%rax	Return value

break becomes a jump to after the switch (.L2)!


Jump tables

 Definition: An array where entry i is the address of the code segment to run when the switch variable equals i

Switch statement

```
switch(x) {
  case 0:
    Block 0
  case 1:
    Block 1
  case n-1:
    Block n-1
```

Jump table (data in memory)



Targ n-1:

Approx. translation:

```
goto *target;
```

- Register %rdi holds the switch variable x
- jtab is the address of the jump table

Q1: which table entry holds the address of the next instruction? The xth (or %rdith)

Q2: what is the memory address of that entry? jtab + %rdi*8


Q3: what is the address of the next instruction to execute? **M[jtab + %rdi*8]**

Code Block

n–1

Jump table for our example

jtab:



Putting it all Together

Jump table (addresses of code blocks)

```
long switch_fun (...)
{
    switch(x) {
        // cases 0,1,2,7,8
        // and default
    }
    w += 5;
    return w;
}
```

%rdi	Argument x
%rsi	Argument y
%rdx	Argument z
%rcx	Argument w
%rax	Return value

```
.L4:
    .quad    .L7 # x=0
    .quad    .L6 # x=1
    .quad    .L5 # x=2
    .quad    .L8 # x=3
    .quad    .L8 # x=4
    .quad    .L8 # x=5
    .quad    .L8 # x=6
    .quad    .L8 # x=6
    .quad    .L3 # x=7
    .quad    .L3 # x=8
```

```
switch_fun:
```

Indirect jump: look up address in memory; jump there

```
long switch fun
 (long x, long y, long z, long w) {
 switch(x) {
 case 0:
     W += V;
     break;
 case 1:
     W = V;
      /* FALL THROUGH */
 case 2:
     W += z;
     break;
  /* MISSING CASES */
 case 7:
 case 8: /* MULTIPLE CASES */
     w = z;
      break;
 default:
     w = 2;
      break;
 w += 5;
 return w;
```

Full assembly code for our example

```
switch fun:
         $8, %rdi
  cmpq
  jа
       .L8
         *.L4(,%rdi,8)
  jmp
.L4:
  .quad .L7
  .quad .L6
  .quad .L5
  .quad .L8
  .quad .L8
  .quad .L8
  .quad .L8
  .quad .L3
         .L3
  .quad
.L7:
        %rsi, %rcx
  addq
  jmp
         .L2
```

```
.L6:
  subq %rsi, %rcx
 # FALL THROUGH
.L5:
  addq %rdx, %rcx
  jmp
         .L2
.L3:
         %rdx, %rcx
  subq
  jmp
         .L2
.L8:
         $2, %ecx
  movl
  jmp
         .L2
.L2:
  leaq
         5(%rcx), %rax
  ret
```

Another Jump Table Example: starting with assembly

- QUIZ: find the address of the jump table and code blocks
 - linux> objdump -d prog
 - The jump table starts at address <a>0x400668
 - The default code block is at address 0x40055c

```
0000000000400528 <switch eg>:
400528:
           48 89 d1
                                           %rdx,%rcx
                                   mov
         48 83 ff 06
                                           $0x6,%rdi
40052b:
                                   cmp
40052f:
         77 2b
                                           \frac{40055c}{\text{<switch eg+0x34>}}
                                   jа
 400531 ff 24 fd 68 06 40 00
                                          *0x400668(,%rdi,8)
                                   jmpg
```

Note: these are hex values (memory addresses for instructions) objdump does not put 0x in front of instruction addresses when it disassembles

How would you find the address of the other code blocks?

Object code: Jump Table

- Jump table
 - Doesn't show up in disassembled code
 - Can inspect using GDB: examine data starting at address
 0x400668


```
gdb prog (gdb) x/7xg 0x400668
```


- Examine 7 hexadecimal format "giant words" (8-bytes each)
- Use command "help x" to get format documentation

0x400668:

0x000000000040055c
0x0000000000400538
0x0000000000400540
0x000000000040054a
0x000000000040055c
0x0000000000400553

 $0 \times 0000000000400553$





How can you see the code for each one of the target code blocks?

Object code: Disassemble targets

```
400538: 48 89 f0
                               %rsi,%rax
                          mov
40053b: 48 Of af c2
                          imul %rdx,%rax
40053f: c3
                          retq
400540: 48 89 f0
                          mov %rsi,%rax
400543: 48 99
                          cqto
400545: 48 f7 f9
                          idiv %rcx
400548: eb 05
                          jmp 40054f < \text{switch eg} + 0x27 >
40054a: b8 01 00 00 00
                          mov $0x1, %eax
40054f: 48 01 c8
                          add %rcx,%rax
400552: c3
                          reta
400553: b8 01 00 00 00
                          mov $0x1, %eax
400558: 48 29 d0
                          sub
                               %rdx,%rax
40055b: c3
                          retq
40055c: b8 02 00 00 00
                          mov $0x2, %eax
400561: c3
                          retq
```

```
.section
          .rodata
  .align 8
.L4:
 .quad .L8 \# x = 0
        .L3 \# x = 1
 .quad
        .L5 \# x = 2
 .quad
        .19 # x = 3
  .quad
  .quad
        .L8 \# x = 4
  .quad
        .L7 # x = 5
          .L7 \# x = 6
  .quad
```

0x40055c

0x400538

0x400540

0x40054a =

0x40055c

0x400553

0x400553

```
linux> gdb prog
(gdb) disassemble 0x400538,0x400562
```

Object code: Disassemble targets

```
400538: 48 89 f0
                                              %rsi,%rax
                                          mov
                 40053b: 48 Of af c2
                                          imul %rdx,%rax
                 40053f: c3
                                          retq
                 400540: 48 89 f0
                                          mov %rsi,%rax
0x40055c
                 400543: 48 99
                                          cqto
0x400538
                 400545: 48 f7 f9
                                          idiv %rcx
0x400540
                 400548: eb 05
                                          jmp 40054f < \text{switch eg} + 0x27 >
0x40054a =
                40054a: b8 01 00 00 00
                                          mov $0x1, %eax
0x40055c
                 40054f: 48 01 c8
                                          add %rcx,%rax
0x400553
                 400552: c3
                                          retq
0x400553
                400553: b8 01 00 00 00
                                          mov $0x1, %eax
                 400558: 48 29 d0
                                          sub %rdx,%rax
                 40055b: c3
                                          retq
                 40055c: b8 02 00 00 00
                                          mov $0x2, %eax
                 400561: c3
                                          retq
```

Object code: Memory View

